


ก่อนเริ่ มอ่าน
Ebook นี้ รวม 30 สูตรที่คนทำงานใช้บ่อยที่สุด ทั้งเจ้าของร้ าน พนักงานออฟฟิศ HR ฝ่ ายจัดซื้อ ครู และคนที่อยากจัดการเงินในบ้าน ทุก

สูตรมีโจทย์จากงานจริ ง ตัวอย่างตาราง วิธีอ่านสูตร จุดที่คนมักพลาด และเคล็ดลับ

อ่านภาพตัวอย่างยังไง

แถบด้านบนตาราง = แถบสูตรของ Excel บอกว่าช่องไหนใส่สูตรอะไร
ช่องที่มีกรอบสีในตาราง = ช่วงข้อมูลที่สูตรอ้างถึง สีตรงกับตัวอักษรในสูตร

ช่องที่มีกรอบหนา = ช่องที่ใส่สูตร ตัวเลขที่เห็นคือผลลัพธ์จริ งที่ Excel คำนวณ

วิธีฝึ กให้ได้ผลเร็ว

1. เปิด Excel ไฟล์เปล่า พิมพ์ตารางตัวอย่างตามภาพ (ใช้เวลาไม่ถึง 2 นาที)
2. พิมพ์สูตรเองทีละตัวอักษร อย่าคัดลอก มือจะจำรูปแบบได้เร็วกว่า

3. ลองเปลี่ยนตัวเลขในตาราง ดูว่าผลลัพธ์เปลี่ยนตามไหม
4. ลองทำให้ผิดตาม "จุดที่มักพลาด" จะจำได้ว่าอาการแบบไหนแก้ยังไง

ใช้กับโปรแกรมไหนได้บ้าง
ทุกสูตรใช้ได้ใน Excel 2016 ขึ้นไป, Microsoft 365, Google Sheets และ WPS ยกเว้น BAHTTEXT ที่ Google Sheets ยังไม่มี

ไม่มีสูตรที่ต้องใช้ Excel รุ่ นใหม่เท่านั้น

ถ้าเครื่ องของคุณใช้เครื่ องหมาย ; คั่นในสูตรแทน , (บางเครื่ องที่ตั้งภาษาเป็ นยุโรป) ให้เปลี่ยน , เป็ น ; ทุกที่

ไม่อยากสร้ างเอง?

ท้ายทุกสูตรมีลิงก์ไปยังไฟล์สำเร็จรูปของสูตรสำเร็จ ที่ใช้สูตรนั้นในงานจริ ง เปิดแล้วกรอกข้อมูลได้เลย ดูไฟล์ทั้งหมดได้ที่

1 SUM

โจทย์  จดรายจ่ายในบ้านทั้งสัปดาห์ไว้ อยากรู้ ว่ารวมแล้วจ่ายไปเท่าไร ไม่ต้องกดเครื่ องคิดเลขทีละบรรทัด

รูปแบบ =SUM(ช่วงตัวเลข)

B7 fx =SUM(B2:B6)

A B

1 รายการ จำนวนเงิน

2 ค่ากับข้าว 1,250 ฿

3 ค่าน้ำ มันรถ 800 ฿

4 ค่าไฟ 1,420 ฿

5 ค่าโทรศัพท์ 399 ฿

6 ค่าขนมลูก 300 ฿

7 รวม 4,169 ฿

B2:B6 ช่วงตัวเลขที่จะรวม

อ่านสูตรนี้ ยังไง

คลิกช่องที่จะใส่ผลรวม พิมพ์ =SUM( แล้วลากเมาส์คลุมตัวเลขทั้งหมด ปิดวงเล็บ กด Enter
ทางลัด: คลิกช่องใต้ตัวเลข แล้วกด Alt + = Excel จะเลือกช่วงให้เอง

⚠ จุดที่มักพลาด
ลากคลุมช่อง "รวม" ของตัวเองเข้าไปด้วย จะเกิดการอ้างอิงวน

(Circular Reference) และได้ 0

💡 เคล็ดลับ
เพิ่มแถวรายการ "ระหว่าง" ช่วงเดิม สูตรจะขยายช่วงให้เอง แต่

ถ้าเพิ่มต่อท้ายใต้ช่วง ต้องแก้ช่วงเอง

ภาคที่ 1 · พื้นฐานที่ใช้ทุกวัน

รวมตัวเลขทั้งคอลัมน์ในคลิกเดียว

ใช้จริ งในไฟล์สำเร็จรูป รายรั บ-รายจ่ายรายเดือน sutsamret.com/templates/income-expense-monthly →

2 AVERAGE · MAX · MIN

โจทย์  ร้ านขายได้ 5 วัน อยากรู้ ว่าเฉลี่ยวันละเท่าไร วันที่ขายดีที่สุดและแย่ที่สุดได้เท่าไร

รูปแบบ =AVERAGE(ช่วง)   =MAX(ช่วง)   =MIN(ช่วง)

B7 fx =AVERAGE(B2:B6) = 4,010 ฿

B8 fx =MAX(B2:B6) = 5,300 ฿

B9 fx =MIN(B2:B6) = 2,850 ฿

A B

1 วัน ยอดขาย

2 จันทร์  3,200 ฿

3 อังคาร 4,100 ฿

4 พุธ 2,850 ฿

5 พฤหัส 5,300 ฿

6 ศุกร์  4,600 ฿

7 เฉลี่ย 4,010 ฿

8 สูงสุด 5,300 ฿

9 ต่ำสุด 2,850 ฿

B2:B6 ยอดขาย 5 วัน

อ่านสูตรนี้ ยังไง

ทั้งสามสูตรใช้ช่วงเดียวกัน ต่างกันแค่ชื่อสูตร
AVERAGE = ผลรวม ÷ จำนวนช่องที่มีตัวเลข (ช่องว่างไม่นับ แต่ช่องที่พิมพ์ 0 นับ)

⚠ จุดที่มักพลาด
วันที่ร้ านปิด ถ้าพิมพ์ 0 ค่าเฉลี่ยจะต่ำลง ถ้าไม่อยากให้นับ ให้

เว้นว่างไว้แทน

💡 เคล็ดลับ
เลือกช่วงตัวเลขด้วยเมาส์ แล้วดูแถบล่างของ Excel จะเห็นค่า

เฉลี่ย จำนวน และผลรวมทันทีโดยไม่ต้องพิมพ์สูตร

ภาคที่ 1 · พื้นฐานที่ใช้ทุกวัน

ค่าเฉลี่ย สูงสุด ต่ำสุด

ใช้จริ งในไฟล์สำเร็จรูป ขายหน้าร้ าน POS ยอดขายรายวัน sutsamret.com/templates/pos-daily-sales →

3 COUNT · COUNTA

โจทย์  เก็บค่าบำรุงชมรมคนละ 200 บาท อยากรู้ ว่ามีสมาชิกทั้งหมดกี่คน และจ่ายแล้วกี่คน

รูปแบบ =COUNTA(ช่วง) นับช่องที่ไม่ว่าง   =COUNT(ช่วง) นับเฉพาะตัวเลข

B7 fx =COUNTA(A2:A6) = 5

B8 fx =COUNT(B2:B6) = 3

A B

1 ชื่อ ค่าบำรุง

2 สมชาย 200

3 สุดา

4 วิชัย 200

5 มาลี

6 อรุณี 200

7 สมาชิกทั้งหมด 5

8 จ่ายแล้ว (คน) 3

A2:A6 รายชื่อ (นับทุกช่องที่ไม่ว่าง) B2:B6 ช่องเงิน (นับเฉพาะตัวเลข)

อ่านสูตรนี้ ยังไง

COUNTA นับทุกอย่างที่ไม่ว่าง ทั้งข้อความและตัวเลข เหมาะกับนับรายชื่อ
COUNT นับเฉพาะตัวเลข ช่องที่เว้นว่างหรื อเป็ นข้อความจะไม่นับ

⚠ จุดที่มักพลาด
ช่องที่ดูเหมือนว่างแต่มีเว้นวรรค (Space) อยู่ COUNTA จะนับ

ด้วย ให้กด Delete ล้างช่องนั้น

💡 เคล็ดลับ
ค้างจ่ายกี่คน = สมาชิกทั้งหมด − จ่ายแล้ว หรื อใช้

=COUNTBLANK(B2:B6) นับช่องว่างตรง ๆ

ภาคที่ 1 · พื้นฐานที่ใช้ทุกวัน

นับว่ามีกี่รายการ

ใช้จริ งในไฟล์สำเร็จรูป เช็คชื่อนักเรี ยน มา / ขาด / ลา / สาย sutsamret.com/templates/attendance-check →

4 ROUND

โจทย์  ซื้อของยกแพ็กมาแบ่งขาย อยากรู้ ต้นทุนต่อชิ้ น แต่หารแล้วได้ทศนิยมยาวเหยียด

รูปแบบ =ROUND(ตัวเลขหรือสูตร, จำนวนทศนิยม)

D2 fx =ROUND(B2/C2,2)

A B C D

1 สินค้า ราคาแพ็ก จำนวนชิ้น ต่อชิ้ น

2 น้ำ ดื่ม 65 ฿ 12 5.42 ฿

3 ไข่ไก่ 112 ฿ 30 3.73 ฿

4 ทิชชู่ 189 ฿ 24 7.88 ฿

B2 ราคาทั้งแพ็ก C2 จำนวนชิ้ นในแพ็ก

อ่านสูตรนี้ ยังไง

65 ÷ 12 = 5.41666… ROUND(…, 2) ปั ดเหลือ 2 ตำแหน่ง ได้ 5.42
เลข 0 = ปั ดเป็ นจำนวนเต็ม เลข -1 = ปั ดเป็ นหลักสิบ เลข -2 = ปั ดเป็ นหลักร้ อย

⚠ จุดที่มักพลาด
กดปุ่ มลดทศนิยมบนแถบเครื่ องมือ แค่ "ซ่อน" ทศนิยม ตัวเลข

จริ งยังยาวเหมือนเดิม พอรวมกันหลายช่องยอดจะคลาดไป
สตางค์สองสตางค์

💡 เคล็ดลับ
ตั้งราคาขายให้ลงท้ายสวย ๆ ใช้ =ROUNDUP(ต้นทุน*1.3,

-1) ปั ดขึ้นเป็ นหลักสิบเสมอ

ภาคที่ 1 · พื้นฐานที่ใช้ทุกวัน

ปั ดเศษทศนิยมให้ตัวเลขสวยและถูกต้อง

ใช้จริ งในไฟล์สำเร็จรูป คำนวณต้นทุนเมนูอาหาร + ตั้งราคาขาย sutsamret.com/templates/restaurant-menu-cost-pricing →

5 $ ล็อกช่อง

โจทย์  ให้ค่าคอม 3% ของยอดขายทุกคน เขียนสูตรแถวแรกแล้วลากลงมา แต่แถวถัดไปกลับได้ 0

รูปแบบ $D$2 = ล็อกทั้งคอลัมน์และแถว (กด F4 ที่ชื่อช่อง)

C3 fx =B3*$D$2

A B C D

1 พนักงาน ยอดขาย ค่าคอม อัตราค่าคอม

2 สุดา 85,000 ฿ 2,550 ฿ 3%

3 วิชัย 62,000 ฿ 1,860 ฿

4 มาลี 91,000 ฿ 2,730 ฿

B3 ยอดขายแถวนี้  (ขยับตามแถว) $D$2 อัตราที่ล็อกไว้ (ไม่ขยับ)

อ่านสูตรนี้ ยังไง

ถ้าเขียน =B2*D2 แล้วลากลง แถวถัดไปจะกลายเป็ น =B3*D3 ซึ่ง D3 ว่าง จึงได้ 0
ใส่ $ หน้าตัวอักษรและตัวเลข ($D$2) ช่องนั้นจะไม่ขยับเวลาลากหรื อคัดลอก

⚠ จุดที่มักพลาด
ล็อกผิดช่อง เช่น ล็อก $B$2 แทน ทุกแถวจะคิดจากยอดขาย

ของคนแรกคนเดียว

💡 เคล็ดลับ
คลิกที่ D2 ในแถบสูตร แล้วกด F4 วนได้ 4 แบบ: $D$2 →
D$2 → $D2 → D2

ภาคที่ 1 · พื้นฐานที่ใช้ทุกวัน

ลากสูตรลงมาแล้วช่องอัตรายังอยู่ที่เดิม

ใช้จริ งในไฟล์สำเร็จรูป ยอดขายเทียบเป้ า + ค่าคอมมิชชั่นทีมขาย sutsamret.com/templates/sales-target-commission →

6 ส่วนลด % และ VAT 7%

โจทย์  เปิดบิลขายเก้าอี้ ลด 10% แล้วต้องบวก VAT 7% อยากให้ไฟล์คิดให้ ไม่ต้องคิดในหัว

รูปแบบ ราคาหลังลด =ราคา*(1-ส่วนลด)   VAT =ROUND(ยอด*7%,2)

D2 fx =B2*(1-C2) = 2,205 ฿

E2 fx =ROUND(D2*7%,2) = 154.35 ฿

F2 fx =D2+E2 = 2,359.35 ฿

A B C D E F

1 สินค้า ราคาเต็ม ส่วนลด หลังลด VAT 7% รวมจ่าย

2 เก้าอี้ สำนักงาน 2,450 ฿ 10% 2,205 ฿ 154.35 ฿ 2,359.35 ฿

3 โต๊ะทำงาน 3,900 ฿ 5% 3,705 ฿ 259.35 ฿ 3,964.35 ฿

B2 ราคาเต็ม C2 ส่วนลด % D2 ยอดหลังลด

อ่านสูตรนี้ ยังไง

ลด 10% แปลว่าจ่าย 90% จึงคูณ (1-10%) ได้ในขั้นเดียว
VAT ปั ด 2 ตำแหน่งทุกครั้ ง ยอดในบิลจะตรงกับที่คิดด้วยมือ

⚠ จุดที่มักพลาด
พิมพ์ส่วนลดเป็ น 10 แทน 10% สูตรจะคิดเป็ น 1,000% ให้
พิมพ์ 10% หรื อ 0.1

💡 เคล็ดลับ
ถ้าราคาที่มีรวม VAT แล้ว หาราคาก่อน VAT ด้วย =ราคา
รวม/1.07

ภาคที่ 1 · พื้นฐานที่ใช้ทุกวัน

ราคาหลังลด ภาษี และยอดที่ลูกค้าต้องจ่าย

ใช้จริ งในไฟล์สำเร็จรูป ใบกำกับภาษีเต็มรูป + ทะเบียนเลขที่รั นอัตโนมัติ sutsamret.com/templates/tax-invoice-register →

7 IF

โจทย์  ครูตรวจงานนักเรี ยน ได้ 50 คะแนนขึ้นไปถือว่าผ่าน อยากให้ช่องผลขึ้นคำว่าผ่าน/ไม่ผ่านเอง

รูปแบบ =IF(เงื่อนไข, ค่าถ้าจริ ง, ค่าถ้าไม่จริ ง)

C2 fx =IF(B2>=50,"ผ่าน","ไม่ผ่าน")

A B C

1 นักเรี ยน คะแนน ผล

2 ด.ช.ก้อง 72 ผ่าน

3 ด.ญ.ใบเตย 45 ไม่ผ่าน

4 ด.ช.ภูมิ 50 ผ่าน

5 ด.ญ.มิ้ นท์ 88 ผ่าน

6 ด.ช.ต้น 31 ไม่ผ่าน

B2 คะแนนที่นำไปเทียบกับ 50

อ่านสูตรนี้ ยังไง

อ่านว่า: ถ้า B2 มากกว่าหรื อเท่ากับ 50 ให้เขียน "ผ่าน" ไม่อย่างนั้นเขียน "ไม่ผ่าน"
เครื่ องหมายเทียบ: = <>(ไม่เท่ากับ) > < >= <=

⚠ จุดที่มักพลาด
ลืมใส่เครื่ องหมายคำพูดรอบข้อความ เช่น ผ่าน ต้องเป็ น

"ผ่าน" ไม่อย่างนั้นได้ #NAME?

💡 เคล็ดลับ
ให้คำตอบเป็ นตัวเลขก็ได้ เช่น =IF(B2>=50,100,0) ให้ 100

บาทคนที่ผ่าน

ภาคที่ 2 · ให้ Excel ตัดสินใจแทน

ถ้า…ให้ตอบอย่างหนึ่ง ถ้าไม่ใช่ให้ตอบอีกอย่าง

ใช้จริ งในไฟล์สำเร็จรูป ติดตามการส่งงานนักเรี ยน + รวมคะแนนเก็บ sutsamret.com/templates/homework-tracker →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 10

https://sutsamret.com/templates/homework-tracker?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


8 IF ซ้อน ตัดเกรด

โจทย์  ต้องตัดเกรด 40 คน 80 ขึ้นไป A, 70 B, 60 C, 50 D ต่ำกว่านั้น F

รูปแบบ =IF(B2>=80,"A",IF(B2>=70,"B",IF(B2>=60,"C",IF(B2>=50,"D","F"))))

C3 fx =IF(B3>=80,"A",IF(B3>=70,"B",IF(B3>=60,"C",IF(B3>=50,"D","F"))))

A B C

1 นักเรี ยน คะแนน เกรด

2 ด.ช.ก้อง 85 A

3 ด.ญ.ใบเตย 74 B

4 ด.ช.ภูมิ 61 C

5 ด.ญ.มิ้ นท์ 52 D

6 ด.ช.ต้น 38 F

B3 คะแนนที่ตรวจทีละขั้น

อ่านสูตรนี้ ยังไง

Excel ตรวจจากซ้ายไปขวา เจอเงื่อนไขแรกที่จริ งก็หยุด 74 ไม่ถึง 80 แต่ถึง 70 จึงได้ B
จำนวนวงเล็บปิดท้าย = จำนวน IF (สูตรนี้ มี 4 IF จึงปิด 4 วงเล็บ)

⚠ จุดที่มักพลาด
เรี ยงเงื่อนไขจากน้อยไปมาก เช่น ตรวจ >=50 ก่อน ทุกคนที่

ได้ 50 ขึ้นไปจะได้ D หมด ต้องเริ่ มจากเกณฑ์สูงสุดก่อน

💡 เคล็ดลับ
ถ้าเกณฑ์มีหลายขั้นหรื อเปลี่ยนบ่อย ใช้ตารางเกณฑ์ +
VLOOKUP แบบประมาณค่า (สูตรที่ 17) แก้ง่ายกว่า

ภาคที่ 2 · ให้ Excel ตัดสินใจแทน

หลายเงื่อนไขเรี ยงกัน A B C D F

ใช้จริ งในไฟล์สำเร็จรูป บันทึกคะแนน + ตัดเกรดนักเรี ยนอัตโนมัติ sutsamret.com/templates/student-grades →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 11

https://sutsamret.com/templates/student-grades?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


9 AND · OR

โจทย์  ให้โบนัสเฉพาะคนที่ยอดขายถึง 50,000 "และ" ไม่ขาดงานเลยในเดือนนั้น

รูปแบบ AND(เงื่อนไข1, เงื่อนไข2) จริงเมื่อจริ งทุกข้อ   OR(…) จริงเมื่อจริ งข้อใดข้อหนึ่ง

D3 fx =IF(AND(B3>=50000,C3=0),"ได้","ไม่ได้")

A B C D

1 พนักงาน ยอดขาย วันขาดงาน โบนัส

2 สุดา 62,000 ฿ 0 ได้

3 วิชัย 71,000 ฿ 1 ไม่ได้

4 มาลี 48,000 ฿ 0 ไม่ได้

5 ธนพล 55,000 ฿ 0 ได้

B3 ยอดขายถึง 50,000 ไหม C3 ขาดงาน 0 วันไหม

อ่านสูตรนี้ ยังไง

วิชัยยอดขายถึงแต่ขาดงาน 1 วัน AND ต้องผ่านทั้งสองข้อ จึงไม่ได้
ถ้าเปลี่ยนเป็ น OR คนที่ผ่านข้อใดข้อหนึ่งก็ได้โบนัส

⚠ จุดที่มักพลาด
เขียน =IF(B2>=50000 AND C2=0, …) แบบภาษา

อังกฤษ Excel ไม่รู้ จัก ต้องใช้ AND( , ) ครอบ

💡 เคล็ดลับ
ผสมกันได้ เช่น

=IF(OR(ยอด>=80000,AND(ยอด>=50000,ขาด=0)),"ได้","ไม่

ได้")

ภาคที่ 2 · ให้ Excel ตัดสินใจแทน

หลายเงื่อนไขพร้ อมกันใน IF เดียว

ใช้จริ งในไฟล์สำเร็จรูป ประเมินผลงานพนักงาน KPI + สมรรถนะ sutsamret.com/templates/kpi-performance-review →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 12

https://sutsamret.com/templates/kpi-performance-review?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


10 IFERROR

โจทย์  คิดต้นทุนต่อจานของแต่ละเมนู แต่เมนูที่วันนี้ ยังขายไม่ได้ (0 จาน) ขึ้ น #DIV/0! น่าเกลียด

รูปแบบ =IFERROR(สูตรเดิม, ค่าที่จะแสดงแทนถ้าผิดพลาด)

D3 fx =IFERROR(B3/C3,0)

A B C D

1 เมนู ต้นทุนรวม ขายได้ (จาน) ต้นทุน/จาน

2 ข้าวผัด 900 ฿ 30 30 ฿

3 ต้มยำกุ้ ง 1,200 ฿ 0 0 ฿

4 กะเพราไก่ 750 ฿ 25 30 ฿

B3 ต้นทุนรวม C3 ขายได้ 0 จาน (หารไม่ได้)

อ่านสูตรนี้ ยังไง

1,200 ÷ 0 หารไม่ได้ ปกติขึ้ น #DIV/0! IFERROR จับไว้แล้วแสดง 0 แทน
ใช้ครอบ VLOOKUP ได้ด้วย เช่น =IFERROR(VLOOKUP(…),"ไม่พบรหัส")

⚠ จุดที่มักพลาด
ครอบ IFERROR ทุกสูตรตั้งแต่แรก ทำให้ไม่เห็นว่าสูตรผิดจริ ง

ให้เขียนสูตรจนถูกก่อน แล้วค่อยครอบ

💡 เคล็ดลับ
อยากให้ช่องว่างแทน 0 ใช้ =IFERROR(B3/C3,"")

ภาคที่ 2 · ให้ Excel ตัดสินใจแทน

ซ่อน #DIV/0! #N/A ให้ตารางดูสะอาด

ใช้จริ งในไฟล์สำเร็จรูป คำนวณต้นทุนเมนูอาหาร + ตั้งราคาขาย sutsamret.com/templates/restaurant-menu-cost-pricing →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 13

https://sutsamret.com/templates/restaurant-menu-cost-pricing?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


11 COUNTIF

โจทย์  บันทึกการเข้างานทั้งทีม อยากรู้ ว่าเดือนนี้ มีคนมาสายกี่ครั้ ง

รูปแบบ =COUNTIF(ช่วงที่จะดู, สิ่งที่ต้องการนับ)

B8 fx =COUNTIF(B2:B7,"สาย")

A B

1 ชื่อ สถานะ

2 สมชาย มา

3 สุดา สาย

4 วิชัย มา

5 มาลี สาย

6 อรุณี ลา

7 ธนพล สาย

8 มาสาย (ครั้ ง) 3

B2:B7 ช่วงสถานะที่จะนับ

อ่านสูตรนี้ ยังไง

นับว่าใน B2:B7 มีคำว่า "สาย" กี่ช่อง ได้ 3
เงื่อนไขตัวเลขใส่ในเครื่ องหมายคำพูด เช่น ">=50" หรื อ "<>0"

⚠ จุดที่มักพลาด
สะกดไม่ตรง เช่น พิมพ์ "สาย " มีเว้นวรรคท้าย จะไม่ถูกนับ ทำ

Dropdown ให้เลือกแทนการพิมพ์

💡 เคล็ดลับ
หารายการซ้ำ : =COUNTIF($A$2:$A$100,A2)>1 ได้

TRUE ที่แถวซ้ำ

ภาคที่ 3 · สรุปยอดตามเงื่อนไข

นับเฉพาะรายการที่ตรงเงื่อนไข

ใช้จริ งในไฟล์สำเร็จรูป เช็คชื่อนักเรี ยน มา / ขาด / ลา / สาย sutsamret.com/templates/attendance-check →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 14

https://sutsamret.com/templates/attendance-check?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


12 SUMIF

โจทย์  จดรายจ่ายทุกอย่างปนกัน อยากรู้ ว่าเดือนนี้ หมดไปกับค่าอาหารเท่าไร

รูปแบบ =SUMIF(ช่วงที่จะเช็ก, เงื่อนไข, ช่วงตัวเลขที่จะรวม)

C7 fx =SUMIF(B2:B6,"อาหาร",C2:C6)

A B C

1 รายการ หมวด จำนวนเงิน

2 กาแฟ อาหาร 65 ฿

3 เติมน้ำ มัน เดินทาง 800 ฿

4 ข้าวเที่ยง อาหาร 120 ฿

5 ทางด่วน เดินทาง 100 ฿

6 ข้าวเย็น อาหาร 250 ฿

7 รวมหมวดอาหาร 435 ฿

B2:B6 ช่วงหมวดที่จะเช็ก C2:C6 ตัวเลขที่จะรวม

อ่านสูตรนี้ ยังไง

Excel ไล่ดูทีละแถว แถวไหนหมวดเป็ น "อาหาร" ก็เอาเงินแถวนั้นมารวม: 65 + 120 + 250 = 435
ใส่ชื่อหมวดไว้ในช่อง แล้วอ้างช่องแทนการพิมพ์ในสูตร จะเปลี่ยนหมวดได้ทันที

⚠ จุดที่มักพลาด
ช่วงที่เช็กกับช่วงที่รวมยาวไม่เท่ากัน (B2:B6 กับ C2:C9) ผลจะ

ผิดโดยไม่มีอะไรเตือน

💡 เคล็ดลับ
ทำตารางสรุปทุกหมวด: เขียนชื่อหมวดเรี ยงลงมา แล้วใช้

=SUMIF($B$2:$B$500,E2,$C$2:$C$500) ลากลง

ภาคที่ 3 · สรุปยอดตามเงื่อนไข

รวมเฉพาะหมวดที่ต้องการ

ใช้จริ งในไฟล์สำเร็จรูป รายรั บ-รายจ่ายรายเดือน sutsamret.com/templates/income-expense-monthly →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 15

https://sutsamret.com/templates/income-expense-monthly?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


13 COUNTIFS

โจทย์  แม่ค้าออนไลน์อยากรู้ ว่าออเดอร์ จาก Shopee ที่ยัง "รอส่ง" อยู่มีกี่ออเดอร์

รูปแบบ =COUNTIFS(ช่วง1, เงื่อนไข1, ช่วง2, เงื่อนไข2, …)

C8 fx =COUNTIFS(B2:B7,"Shopee",C2:C7,"รอส่ง")

A B C

1 ออเดอร์ ช่องทาง สถานะ

2 #1021 Shopee ส่งแล้ว

3 #1022 Lazada รอส่ง

4 #1023 Shopee รอส่ง

5 #1024 Facebook รอส่ง

6 #1025 Shopee รอส่ง

7 #1026 Shopee ส่งแล้ว

8 Shopee รอส่ง 2

B2:B7 เงื่อนไข 1: ช่องทาง Shopee C2:C7 เงื่อนไข 2: สถานะรอส่ง

อ่านสูตรนี้ ยังไง

นับเฉพาะแถวที่ผ่านทุกเงื่อนไข: #1023 และ #1025 ได้ 2
เพิ่มเงื่อนไขได้เรื่ อย ๆ เป็ นคู่ ช่วง, เงื่อนไข

⚠ จุดที่มักพลาด
ช่วงแต่ละคู่ต้องยาวเท่ากันและเริ่ มแถวเดียวกัน ไม่อย่างนั้นได้

#VALUE!

💡 เคล็ดลับ
นับช่วงวันที่: =COUNTIFS(วัน

ที่,">="&DATE(2026,10,1),วันที่,"<"&DATE(2026,11,1))
= ออเดอร์ เดือน ต.ค.

ภาคที่ 3 · สรุปยอดตามเงื่อนไข

นับเมื่อเข้าเงื่อนไขหลายข้อพร้ อมกัน

ใช้จริ งในไฟล์สำเร็จรูป ออเดอร์แม่ค้าออนไลน์ + กำไรสุทธิ sutsamret.com/templates/online-shop-orders-profit →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 16

14 SUMIFS

โจทย์  ร้ านกาแฟ 2 สาขา อยากรู้ ยอดขาย "กาแฟ" ของ "สาขาบางนา" อย่างเดียว

รูปแบบ =SUMIFS(ช่วงที่จะรวม, ช่วง1, เงื่อนไข1, ช่วง2, เงื่อนไข2, …)

C7 fx =SUMIFS(C2:C6,A2:A6,"บางนา",B2:B6,"กาแฟ")

A B C

1 สาขา หมวด ยอดขาย

2 บางนา กาแฟ 3,500 ฿

3 บางนา เบเกอรี่  1,200 ฿

4 สีลม กาแฟ 4,200 ฿

5 บางนา กาแฟ 2,800 ฿

6 สีลม เบเกอรี่  900 ฿

7 กาแฟ บางนา 6,300 ฿

C2:C6 ยอดที่จะรวม (ใส่ก่อน) A2:A6 เงื่อนไข: สาขาบางนา B2:B6 เงื่อนไข: หมวดกาแฟ

อ่านสูตรนี้ ยังไง

SUMIFS ใส่ช่วงที่จะรวมเป็ นอันดับแรก (ต่างจาก SUMIF ที่ใส่ท้าย)
แถวที่ผ่านทั้งสองเงื่อนไขคือ 3,500 และ 2,800 รวม 6,300

⚠ จุดที่มักพลาด
สลับลำดับแบบ SUMIF โดยเอาช่วงรวมไว้ท้าย สูตรจะรวมผิด

คอลัมน์หรื อ error

💡 เคล็ดลับ
ใช้ SUMIFS แทน SUMIF ไปเลยก็ได้ แม้มีเงื่อนไขเดียว จะได้

จำรูปแบบเดียว

ภาคที่ 3 · สรุปยอดตามเงื่อนไข

รวมยอดเมื่อเข้าหลายเงื่อนไข

ใช้จริ งในไฟล์สำเร็จรูป ขายหน้าร้ าน POS ยอดขายรายวัน sutsamret.com/templates/pos-daily-sales →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 17

https://sutsamret.com/templates/pos-daily-sales?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


15 SUMPRODUCT

โจทย์  อยากรู้ มูลค่าสต๊อกทั้งร้ าน = จำนวนคงเหลือ × ต้นทุนต่อชิ้ น ของทุกสินค้ารวมกัน

รูปแบบ =SUMPRODUCT(ช่วงจำนวน, ช่วงราคา)

C6 fx =SUMPRODUCT(B2:B5,C2:C5)

A B C

1 สินค้า คงเหลือ ต้นทุน/ชิ้ น

2 น้ำ ดื่ม 120 5.5 ฿

3 บะหมี่กึ่งฯ 200 6 ฿

4 นมกล่อง 48 12 ฿

5 ขนมปั ง 30 22 ฿

6 มูลค่าสต๊อก 3,096 ฿

B2:B5 จำนวนคงเหลือ C2:C5 ต้นทุนต่อชิ้ น

อ่านสูตรนี้ ยังไง

คูณกันทีละแถว (120×5.5, 200×6, …) แล้วรวมทั้งหมด ไม่ต้องทำคอลัมน์ผลคูณเพิ่ม
ช่วงทั้งสองต้องยาวเท่ากัน

⚠ จุดที่มักพลาด
มีช่องที่เป็ นข้อความ เช่น "-" ในช่วง SUMPRODUCT จะนับ

เป็ น 0 ให้ตรวจว่าเป็ นตัวเลขจริ ง

💡 เคล็ดลับ
SUMPRODUCT ใช้แทนสูตรเงื่อนไขซับซ้อนได้ และทำงานใน

Excel 2016 / Google Sheets / WPS ทุกตัว

ภาคที่ 3 · สรุปยอดตามเงื่อนไข

จำนวน × ราคา ทุกแถว แล้วรวมในสูตรเดียว

ใช้จริ งในไฟล์สำเร็จรูป สต๊อกสินค้า รั บเข้า-จ่ายออก-คงเหลือ sutsamret.com/templates/inventory-stock →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 18

16 VLOOKUP

โจทย์  ทำใบเสนอราคาบ่อย ๆ อยากพิมพ์แค่รหัสสินค้า แล้วให้ราคาขึ้นเองจากตารางสินค้า

รูปแบบ =VLOOKUP(สิ่งที่หา, ตาราง, คอลัมน์ที่, FALSE)

F2 fx =VLOOKUP(E2,A2:C5,3,FALSE)

A B C D E F

1 รหัส สินค้า ราคา รหัสที่ขาย ราคา

2 P01 กระดาษ A4 115 ฿ P03 1,890 ฿

3 P02 แฟ้ ม 3 นิ้ ว 65 ฿

4 P03 หมึกพิมพ์ 1,890 ฿

5 P04 เมาส์ไร้ สาย 390 ฿

E2 รหัสที่ต้องการหา A2:C5 ตารางสินค้า (รหัสต้องอยู่คอลัมน์แรก)

อ่านสูตรนี้ ยังไง

หา P03 ในคอลัมน์แรกของตาราง เจอแล้วเลื่อนไปคอลัมน์ที่ 3 (ราคา) ได้ 1,890
FALSE = ต้องตรงทุกตัวอักษร ใช้ทุกครั้ งที่หารหัสหรื อชื่อ

⚠ จุดที่มักพลาด
ลืมใส่ FALSE จะได้ราคาผิดแบบไม่มี error และลากสูตรลงโดย

ไม่ล็อกตาราง ($A$2:$C$5) แถวล่าง ๆ จะหาไม่เจอ

💡 เคล็ดลับ
ครอบด้วย IFERROR(…,"ไม่พบรหัส") จะรู้ ทันทีว่าพิมพ์รหัสผิด

ภาคที่ 4 · ค้นหาข้อมูลจากตาราง

พิมพ์รหัสสินค้า แล้วดึงราคามาให้เอง

ใช้จริ งในไฟล์สำเร็จรูป ใบเสนอราคา / ใบแจ้งหนี้  / ใบเสร็จรั บเงิน sutsamret.com/templates/quotation-invoice →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 19

17 VLOOKUP แบบช่วง

โจทย์  ค่าคอมเป็ นขั้นบันได: ยอด 50,000 ขึ้นไปได้ 2% 100,000 ขึ้นไป 3% 200,000 ขึ้นไป 5%

รูปแบบ =VLOOKUP(ยอด, ตารางขั้น, 2, TRUE)

F2 fx =VLOOKUP(E2,$A$2:$B$5,2,TRUE)

A B C D E F

1 ยอดตั้งแต่ อัตรา พนักงาน ยอดขาย อัตราที่ได้

2 0 ฿ 0% สุดา 132,000 ฿ 3%

3 50,000 ฿ 2% วิชัย 48,000 ฿ 0%

4 100,000 ฿ 3% มาลี 215,000 ฿ 5%

5 200,000 ฿ 5%

E2 ยอดขายของสุดา $A$2:$B$5 ตารางขั้น (เรี ยงน้อยไปมาก)

อ่านสูตรนี้ ยังไง

TRUE = หาขั้นที่ใกล้ที่สุดที่ไม่เกินค่าที่หา 132,000 อยู่ในขั้น 100,000 จึงได้ 3%
แก้อัตราหรื อเพิ่มขั้นในตารางได้เลย ไม่ต้องแก้สูตร

⚠ จุดที่มักพลาด
ตารางขั้นไม่ได้เรี ยงจากน้อยไปมาก ผลจะผิดโดยไม่มี error

และต้องมีขั้นเริ่ มที่ 0 ไม่อย่างนั้นยอดน้อย ๆ ได้ #N/A

💡 เคล็ดลับ
ใช้แบบเดียวกันกับอัตราภาษีขั้นบันได ส่วนลดตามจำนวนซื้อ

หรื อเกรดตามคะแนน

ภาคที่ 4 · ค้นหาข้อมูลจากตาราง

หาอัตราตามขั้น เช่น ค่าคอมตามยอดขาย

ใช้จริ งในไฟล์สำเร็จรูป ยอดขายเทียบเป้ า + ค่าคอมมิชชั่นทีมขาย sutsamret.com/templates/sales-target-commission →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 20

18 INDEX + MATCH

โจทย์  ตารางพนักงานเก็บรหัสไว้คอลัมน์ขวาสุด VLOOKUP หาย้อนไปทางซ้ายไม่ได้

รูปแบบ =INDEX(ช่วงคำตอบ, MATCH(สิ่งที่หา, ช่วงที่หา, 0))

F2 fx =INDEX(A2:A5,MATCH(E2,C2:C5,0))

A B C D E F

1 ชื่อ แผนก รหัส รหัสที่หา ชื่อ

2 สมชาย ใจดี ขาย E001 E003 มาลี ศรี สุข

3 สุดา แก้วใส ขาย E002

4 มาลี ศรี สุข บัญชี E003

5 ธนพล รั กงาน การตลาด E004

A2:A5 ช่วงคำตอบ (ชื่อ) E2 รหัสที่หา C2:C5 ช่วงที่หา (รหัส)

อ่านสูตรนี้ ยังไง

MATCH หาว่า E003 อยู่แถวที่เท่าไรในช่วงรหัส ได้ 3
INDEX ไปหยิบข้อมูลแถวที่ 3 ของช่วงชื่อ ได้ มาลี ศรี สุข

⚠ จุดที่มักพลาด
ลืมเลข 0 ท้าย MATCH จะเป็ นการหาแบบประมาณค่า ได้คำ

ตอบผิด

💡 เคล็ดลับ
เพิ่ม/ลบคอลัมน์ในตารางแล้วสูตรไม่พัง ต่างจาก VLOOKUP

ที่ต้องแก้เลขคอลัมน์

ภาคที่ 4 · ค้นหาข้อมูลจากตาราง

ค้นหาได้ทุกทิศ แม้รหัสอยู่ขวาสุด

ใช้จริ งในไฟล์สำเร็จรูป คำนวณเงินเดือน + ประกันสังคม + ภาษีหัก ณ ที่จ่าย sutsamret.com/templates/payroll-sso-tax →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 21

19 TODAY

โจทย์  มีค่างวดหลายรายการ อยากรู้ ว่าอีกกี่วันถึงวันจ่ายของแต่ละรายการ

รูปแบบ =TODAY()   จำนวนวัน = วันที่ − TODAY()

C2 fx =B2-TODAY()

A B C

1 รายการผ่อน จ่ายครั้ งถัดไป อีกกี่วัน

2 ผ่อนมือถือ 31/10/2026 3

3 ผ่อนรถ 9/11/2026 12

4 บัตรเครดิต 17/11/2026 20

B2 วันครบกำหนด

อ่านสูตรนี้ ยังไง

วันที่ใน Excel คือตัวเลขนับวัน จึงลบกันได้ตรง ๆ ได้เป็ นจำนวนวัน
ตัวเลขในภาพคือวันที่ทำ Ebook นี้  ในไฟล์ของคุณจะเปลี่ยนตามวันที่เปิด

⚠ จุดที่มักพลาด
ผลลัพธ์ขึ้ นเป็ นวันที่แปลก ๆ เช่น 3/1/1900 แปลว่าช่องถูกจัด

รูปแบบเป็ นวันที่ ให้เปลี่ยนเป็ น General หรื อ Number

💡 เคล็ดลับ
กด Ctrl + ; ใส่วันที่วันนี้ แบบไม่เปลี่ยน (เหมาะกับวันที่ทำ

รายการ)

ภาคที่ 5 · วันที่และกำหนดส่ง

วันที่วันนี้  อัปเดตเองทุกครั้ งที่เปิดไฟล์

ใช้จริ งในไฟล์สำเร็จรูป สรุปค่าผ่อนทั้งหมด เดือนนี้ จ่ายเท่าไร ผ่อนหมดเมื่อไร sutsamret.com/templates/installment-tracker →

20 DATEDIF

โจทย์  HR ต้องคิดอายุงานพนักงาน ณ สิ้ นปี  เพื่อดูสิทธิ์พักร้ อนและเงินรางวัลอายุงาน

รูปแบบ =DATEDIF(วันเริ่ ม, วันสิ้ นสุด, "Y")  ใช้ "M" = เดือน  "YM" = เดือนที่เหลือหลังหักปี

D2 fx =DATEDIF(B2,C2,"Y") = 7

E2 fx =DATEDIF(B2,C2,"YM") = 9

A B C D E

1 พนักงาน เริ่ มงาน ณ วันที่ ปี เดือน

2 สมชาย 1/3/2019 31/12/2026 7 9

3 สุดา 15/6/2021 31/12/2026 5 6

4 ธนพล 5/2/2024 31/12/2026 2 10

B2 วันเริ่ มงาน C2 วันที่คิดถึง

อ่านสูตรนี้ ยังไง

"Y" นับปี เต็ม "YM" นับเดือนที่เกินจากปี เต็ม ใช้คู่กันได้ "x ปี  y เดือน"
อยากคิดถึงวันนี้  ใช้ TODAY() แทนช่องวันสิ้ นสุด

⚠ จุดที่มักพลาด
DATEDIF ไม่มีในรายการสูตรที่ Excel แนะนำ ต้องพิมพ์เองให้

ครบ และวันเริ่ มต้องไม่หลังวันสิ้ นสุด ไม่อย่างนั้นได้ #NUM!

💡 เคล็ดลับ
แสดงเป็ นข้อความ: =DATEDIF(B2,C2,"Y")&" ปี

"&DATEDIF(B2,C2,"YM")&" เดือน"

ภาคที่ 5 · วันที่และกำหนดส่ง

อายุ อายุงาน กี่ปี  กี่เดือน

ใช้จริ งในไฟล์สำเร็จรูป บันทึกวันลาพนักงาน + สิทธิ์ลาคงเหลือ sutsamret.com/templates/employee-leave-tracker →

21 วันครบกำหนด

โจทย์  สั่งของจากผู้ ขายหลายเจ้า แต่ละเจ้าให้เครดิตไม่เท่ากัน อยากรู้ ว่าต้องจ่ายวันไหน

รูปแบบ =วันที่ + จำนวนวัน

D2 fx =B2+C2

A B C D

1 ใบสั่งซื้อ วันที่ส่งของ เครดิต (วัน) ครบกำหนดจ่าย

2 PO-101 5/10/2026 30 4/11/2026

3 PO-102 12/10/2026 45 26/11/2026

4 PO-103 20/10/2026 15 4/11/2026

B2 วันที่ได้รั บของ C2 เครดิตเป็ นวัน

อ่านสูตรนี้ ยังไง

บวกจำนวนวันเข้ากับวันที่ได้เลย ช่องผลลัพธ์ต้องจัดรูปแบบเป็ นวันที่

อยากรู้ ว่าเลยกำหนดมาแล้วกี่วัน ใช้ =TODAY()-D2

⚠ จุดที่มักพลาด
วันที่ที่พิมพ์เองบางเครื่ องกลายเป็ นข้อความ (ชิดซ้าย) บวกเลข

ไม่ได้ ได้ #VALUE! ให้พิมพ์ใหม่ตามรูปแบบวันที่ของเครื่ อง

💡 เคล็ดลับ
ใช้ Conditional Formatting ทำแถวที่ครบกำหนดภายใน 7

วันเป็ นสีเหลือง จะไม่ลืมจ่าย

ภาคที่ 5 · วันที่และกำหนดส่ง

วันที่ + เครดิตกี่วัน = วันต้องจ่าย

ใช้จริ งในไฟล์สำเร็จรูป ใบสั่งซื้อ (PO) + ติดตามของค้างรั บ sutsamret.com/templates/purchase-order-tracker →

22 NETWORKDAYS

โจทย์  วางแผนงานโปรเจกต์ อยากรู้ ว่าแต่ละงานใช้วันทำงานจริ งกี่วัน ไม่นับวันหยุด

รูปแบบ =NETWORKDAYS(วันเริ่ ม, วันจบ, [ช่วงวันหยุด])

D2 fx =NETWORKDAYS(B2,C2)

A B C D

1 งาน เริ่ ม เสร็จ วันทำงาน

2 ออกแบบ 2/11/2026 13/11/2026 10

3 ทำต้นแบบ 16/11/2026 4/12/2026 15

4 ทดสอบ 7/12/2026 18/12/2026 10

B2 วันเริ่ ม C2 วันเสร็จ

อ่านสูตรนี้ ยังไง

นับรวมวันเริ่ มและวันจบ ตัดเสาร์ -อาทิตย์ออก 2–13 พ.ย. จึงได้ 10 วัน
ใส่ช่วงวันหยุดนักขัตฤกษ์เป็ นตัวที่ 3 เพื่อตัดวันหยุดราชการออกด้วย

⚠ จุดที่มักพลาด
ลืมใส่วันหยุด เช่น 5 ธ.ค. 10 ธ.ค. วันทำงานจะเกินความจริ ง

💡 เคล็ดลับ
ทำงานวันเสาร์ ด้วย ใช้ NETWORKDAYS.INTL(เริ่ ม, จบ, 11)

= หยุดเฉพาะอาทิตย์

ภาคที่ 5 · วันที่และกำหนดส่ง

นับเฉพาะวันทำงาน ข้ามเสาร์ -อาทิตย์

ใช้จริ งในไฟล์สำเร็จรูป
แผนงานโปรเจกต์ Gantt Chart + ติดตามความคืบ
หน้า

sutsamret.com/templates/project-gantt-chart →

23 EOMONTH

โจทย์  หอพักให้จ่ายค่าเช่าภายในสิ้ นเดือน อยากให้ใบแจ้งหนี้ ใส่วันครบกำหนดเอง

รูปแบบ =EOMONTH(วันที่, 0) สิ้ นเดือนนี้    =EOMONTH(วันที่, 1) สิ้ นเดือนหน้า

C2 fx =EOMONTH(B2,0)

A B C

1 ห้อง วันที่ออกบิล ครบกำหนด (สิ้ นเดือน)

2 101 5/2/2026 28/2/2026

3 102 5/4/2026 30/4/2026

4 103 5/12/2026 31/12/2026

B2 วันที่ออกบิล

อ่านสูตรนี้ ยังไง

เดือน ก.พ. ได้ 28 เม.ย. ได้ 30 ธ.ค. ได้ 31 คิดให้เองทุกเดือน
เลขตัวที่ 2 คือเลื่อนกี่เดือน: 0 เดือนนี้  1 เดือนหน้า -1 เดือนก่อน

⚠ จุดที่มักพลาด
ผลขึ้นเป็ นตัวเลข 5 หลัก เช่น 46081 ให้จัดรูปแบบช่องเป็ นวัน

ที่

💡 เคล็ดลับ
วันแรกของเดือน = EOMONTH(วันที่,-1)+1 ใช้ทำรายงาน

รายเดือน

ภาคที่ 5 · วันที่และกำหนดส่ง

วันสิ้ นเดือน ไม่ต้องจำว่าเดือนไหนมีกี่วัน

ใช้จริ งในไฟล์สำเร็จรูป หอพัก / ห้องเช่า: ค่าเช่า ค่าน้ำ  ค่าไฟ + ใบแจ้งหนี้  sutsamret.com/templates/dorm-rent-utility-bills →

24 & รวมข้อความ

โจทย์  รายชื่อแขกงานแต่งแยกชื่อกับนามสกุลไว้ อยากได้ชื่อเต็มไปพิมพ์การ์ ดเชิญ

รูปแบบ =A2&" "&B2

D2 fx =A2&B2&" "&C2

A B C D

1 คำนำหน้า ชื่อ นามสกุล ชื่อบนการ์ด

2 คุณ สมชาย ใจดี คุณสมชาย ใจดี

3 คุณ สุดา แก้วใส คุณสุดา แก้วใส

4 ดร. วิชัย มั่นคง ดร.วิชัย มั่นคง

A2 คำนำหน้า B2 ชื่อ C2 นามสกุล

อ่านสูตรนี้ ยังไง

& คือเครื่ องหมายต่อข้อความ " " คือเว้นวรรคหนึ่งช่อง

ต่อข้อความกับตัวเลขก็ได้ เช่น ="ยอด "&C2&" บาท"

⚠ จุดที่มักพลาด
ลืมเว้นวรรค ได้ สมชายใจดี ติดกัน

💡 เคล็ดลับ
ต่อตัวเลขที่มีทศนิยมหรื อวันที่ ให้ครอบด้วย TEXT() (สูตรถัด

ไป) ไม่อย่างนั้นได้ตัวเลขดิบ

ภาคที่ 6 · ข้อความและเอกสาร

ชื่อ + นามสกุล หรื อสร้ างประโยคจากหลายช่อง

ใช้จริ งในไฟล์สำเร็จรูป วางแผนงานแต่ง: งบประมาณ + รายชื่อแขก sutsamret.com/templates/wedding-budget-guests →

25 TEXT

โจทย์  อยากสร้ างข้อความทวงหนี้ ส่ง LINE ลูกค้า มียอดเงินมีลูกน้ำ  และวันครบกำหนด

รูปแบบ =TEXT(ค่า, "รูปแบบ")  เช่น "#,##0.00"  "d/m/yyyy"

D2 fx ="เรี ยน "&A2&" ยอดค้าง "&TEXT(B2,"#,##0.00")&" บาท ครบกำหนด "&TEXT(C2,"d/m/yyyy")

A B C D

1 ลูกค้า ยอดค้าง ครบกำหนด ข้อความแจ้งลูกค้า

2 ร้ านสมใจ 12,500 ฿ 15/11/2026 เรี ยน ร้ านสมใจ ยอดค้าง 12,500.00 บาท ครบกำหนด 15/11/2026

3 ร้ านเจริ ญ 3,780.5 ฿ 20/11/2026 เรี ยน ร้ านเจริ ญ ยอดค้าง 3,780.50 บาท ครบกำหนด 20/11/2026

B2 ยอดเงิน → #,##0.00 C2 วันที่ → d/m/yyyy

อ่านสูตรนี้ ยังไง

ถ้าไม่ใช้ TEXT ยอดจะออกมาเป็ น 12500 และวันที่เป็ นตัวเลข 46341
รูปแบบที่ใช้บ่อย: "#,##0" ไม่มีทศนิยม "0%" เปอร์ เซ็นต์ "dd/mm/yy" วันที่สั้ น

⚠ จุดที่มักพลาด
TEXT ให้ผลเป็ นข้อความ นำไปบวกลบต่อไม่ได้ ใช้ตอนแสดง
ผลเท่านั้น

💡 เคล็ดลับ
คัดลอกช่องข้อความไปวางใน LINE ได้เลย ประหยัดเวลาพิมพ์
ทีละคน

ภาคที่ 6 · ข้อความและเอกสาร

จัดรูปแบบตัวเลขและวันที่ตอนต่อข้อความ

ใช้จริ งในไฟล์สำเร็จรูป ลูกหนี้  / เจ้าหนี้  ติดตามยอดค้างและเครดิต sutsamret.com/templates/receivable-payable-tracker →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 28

https://sutsamret.com/templates/receivable-payable-tracker?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


26 BAHTTEXT

โจทย์  ใบเสร็จและใบกำกับภาษีต้องมีจำนวนเงินเป็ นตัวหนังสือ พิมพ์เองทุกใบเสี่ยงผิด

รูปแบบ =BAHTTEXT(จำนวนเงิน)

B2 fx =BAHTTEXT(A2)

A B

1 จำนวนเงิน ตัวอักษร

2 1,250.5 ฿ หนึ่งพันสองร้ อยห้าสิบบาทห้าสิบสตางค์

3 23,540 ฿ สองหมื่นสามพันห้าร้ อยสี่สิบบาทถ้วน

4 100,000 ฿ หนึ่งแสนบาทถ้วน

A2 จำนวนเงิน

อ่านสูตรนี้ ยังไง

ได้ทั้งบาทและสตางค์ ลงท้าย "ถ้วน" ให้เองเมื่อไม่มีสตางค์
ใส่วงเล็บแบบในเอกสารจริ ง: ="("&BAHTTEXT(A2)&")"

⚠ จุดที่มักพลาด
Google Sheets ยังไม่มี BAHTTEXT (ใช้ได้ใน Excel และ

WPS) ถ้าเปิดใน Google Sheets จะขึ้น #NAME?

💡 เคล็ดลับ
ใช้กับยอดรวมทั้งสิ้ น (หลังบวก VAT) เสมอ ไม่ใช่ยอดก่อนภาษี

ภาคที่ 6 · ข้อความและเอกสาร

เขียนจำนวนเงินเป็ นตัวอักษรไทย

ใช้จริ งในไฟล์สำเร็จรูป ใบเสนอราคา / ใบแจ้งหนี้  / ใบเสร็จรั บเงิน sutsamret.com/templates/quotation-invoice →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 29

27 PMT

โจทย์  จะกู้ ซื้อบ้านหรื อกู้ ส่วนบุคคล อยากรู้ ว่าต้องผ่อนเดือนละเท่าไร ก่อนไปคุยกับธนาคาร

รูปแบบ =PMT(ดอกเบี้ ยต่อปี/12, จำนวนเดือน, -ยอดกู้)

D2 fx =PMT(B2/12,C2,-A2)

A B C D

1 ยอดกู้ ดอกเบี้ ย/ปี จำนวนเดือน ค่างวด/เดือน

2 1,500,000 ฿ 6% 240 10,746.47 ฿

3 300,000 ฿ 5% 60 5,661.37 ฿

B2 ดอกเบี้ ยต่อปี  (หาร 12 เป็ นต่อเดือน) C2 จำนวนงวด A2 ยอดกู้ (ใส่ลบ)

อ่านสูตรนี้ ยังไง

ดอกเบี้ ยและจำนวนงวดต้องเป็ นหน่วยเดียวกัน ผ่อนรายเดือนจึงหารดอกเบี้ ยด้วย 12
ใส่ยอดกู้ติดลบ (-A2) ค่างวดจะออกมาเป็ นบวก

⚠ จุดที่มักพลาด
สินเชื่อรถยนต์ในไทยส่วนใหญ่คิดดอกเบี้ ยคงที่ (Flat rate) ใช้

PMT ไม่ได้ ค่างวด = (ยอดจัด + ยอดจัด×ดอกเบี้ ย×ปี ) ÷

จำนวนงวด

💡 เคล็ดลับ
ดอกเบี้ ยทั้งหมดที่จะจ่าย = ค่างวด × จำนวนงวด − ยอดกู้

ภาคที่ 7 · เงิน ดอกเบี้ ย และอันดับ

ค่างวดผ่อนต่อเดือน (ดอกเบี้ ยลดต้นลดดอก)

ใช้จริ งในไฟล์สำเร็จรูป คำนวณผ่อนบ้าน / รถ / สินเชื่อ พร้ อมตารางผ่อน sutsamret.com/templates/loan-calculator →

28 FV

โจทย์  ตั้งใจออมเดือนละ 3,000 บาท ได้ผลตอบแทนปีละ 5% อีก 10 ปีจะมีเงินเท่าไร

รูปแบบ =FV(ผลตอบแทนต่อปี/12, จำนวนเดือน, -เงินออมต่อเดือน)

D2 fx =FV(B2/12,C2*12,-A2) = 465,846.84 ฿

E2 fx =A2*C2*12 = 360,000 ฿

A B C D E

1 ออม/เดือน ผลตอบแทน/ปี จำนวนปี เงินที่จะมี เงินต้นที่ออม

2 3,000 ฿ 5% 10 465,846.84 ฿ 360,000 ฿

3 5,000 ฿ 4% 20 1,833,873.13 ฿ 1,200,000 ฿

B2 ผลตอบแทนต่อปี C2 จำนวนปี  (×12 เป็ นเดือน) A2 เงินออมต่อเดือน

อ่านสูตรนี้ ยังไง

ออมเอง 360,000 บาท แต่ดอกเบี้ ยทบต้นทำให้มีมากกว่านั้น ส่วนต่างคือดอกผล
ยิ่งออมนาน ส่วนที่มาจากดอกเบี้ ยยิ่งโตเร็ว

⚠ จุดที่มักพลาด
ใส่ 5 แทน 5% จะคิดเป็ นผลตอบแทน 500% ต่อปี

💡 เคล็ดลับ
อยากรู้ ว่าต้องออมเดือนละเท่าไรถึงได้ 1 ล้าน ใช้

=PMT(5%/12, 120, 0, -1000000)

ภาคที่ 7 · เงิน ดอกเบี้ ย และอันดับ

ออมทุกเดือน ดอกเบี้ ยทบต้น จะมีเงินเท่าไร

ใช้จริ งในไฟล์สำเร็จรูป วางแผนเกษียณ + คำนวณดอกเบี้ ยทบต้น sutsamret.com/templates/retirement-compound-interest →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 31

29 RANK

โจทย์  สรุปยอดขายประจำเดือน อยากรู้ ว่าใครได้ที่ 1 2 3 โดยไม่ต้องเรี ยงข้อมูลใหม่

รูปแบบ =RANK(ค่าของแถวนี้ , $ช่วงทั้งหมด$)

C3 fx =RANK(B3,$B$2:$B$6)

A B C

1 พนักงาน ยอดขาย อันดับ

2 สุดา 132,000 ฿ 2

3 วิชัย 48,000 ฿ 5

4 มาลี 215,000 ฿ 1

5 ธนพล 96,000 ฿ 4

6 อรุณี 132,000 ฿ 2

B3 ยอดของแถวนี้  $B$2:$B$6 ยอดทั้งหมด (ล็อก $)

อ่านสูตรนี้ ยังไง

มากสุดได้อันดับ 1 ยอดเท่ากันได้อันดับเดียวกัน (สุดาและอรุณีได้ที่ 2 ร่วม) แล้วข้ามไปที่ 4
ใส่เลข 1 เป็ นตัวที่ 3 เพื่อให้น้อยสุดได้ที่ 1 เช่น เวลาวิ่งหรื อของเสีย

⚠ จุดที่มักพลาด
ไม่ล็อกช่วงด้วย $ พอลากลง ช่วงเลื่อนตาม อันดับจะผิด

💡 เคล็ดลับ
อยากได้ชื่อคนที่ได้อันดับ 1:

=INDEX(A2:A6,MATCH(MAX(B2:B6),B2:B6,0))

ภาคที่ 7 · เงิน ดอกเบี้ ย และอันดับ

จัดอันดับยอดขายหรื อคะแนน

ใช้จริ งในไฟล์สำเร็จรูป ยอดขายเทียบเป้ า + ค่าคอมมิชชั่นทีมขาย sutsamret.com/templates/sales-target-commission →

สูตรสำเร็จ · 30 สูตร Excel ใช้จริ ง 32

https://sutsamret.com/templates/sales-target-commission?utm_source=ebook&utm_medium=pdf&utm_campaign=excel-30-formulas


30 % เติบโต + ยอดสะสม

โจทย์  เจ้าของร้ านอยากเห็นว่ายอดขายแต่ละเดือนโตขึ้นหรื อลดลง และทั้งปีขายไปแล้วเท่าไร

รูปแบบ % เติบโต =(เดือนนี้ -เดือนก่อน)/เดือนก่อน   ยอดสะสม =ยอดสะสมแถวบน+เดือนนี้

C4 fx =(B4-B3)/B3 = -5%

D4 fx =D3+B4 = 383,000 ฿

A B C D

1 เดือน ยอดขาย เทียบเดือนก่อน ยอดสะสม

2 ม.ค. 120,000 ฿ 120,000 ฿

3 ก.พ. 135,000 ฿ 12% 255,000 ฿

4 มี.ค. 128,000 ฿ -5% 383,000 ฿

5 เม.ย. 150,000 ฿ 17% 533,000 ฿

B4 เดือนนี้  B3 เดือนก่อน D3 ยอดสะสมถึงเดือนก่อน

อ่านสูตรนี้ ยังไง

มี.ค. ขายได้น้อยกว่า ก.พ. จึงติดลบ -5% เห็นทันทีว่าต้องหาสาเหตุ
ยอดสะสมเอายอดสะสมแถวบนบวกเดือนนี้  ลากลงไปได้ทั้งปี

⚠ จุดที่มักพลาด
จัดรูปแบบช่องเป็ น % แล้วยังคูณ 100 อีก จะได้ 1,250% แทน

12.5%

💡 เคล็ดลับ
ยอดสะสมอีกแบบ: =SUM($B$2:B2) แล้วลากลง ช่วงจะยืด

เอง

ภาคที่ 7 · เงิน ดอกเบี้ ย และอันดับ

เดือนนี้ โตกี่เปอร์ เซ็นต์ และรวมตั้งแต่ต้นปี เท่าไร

ใช้จริ งในไฟล์สำเร็จรูป บัญชีรายรั บ-รายจ่ายร้ านค้า + งบกำไรขาดทุน sutsamret.com/templates/shop-books-income-tax →

ไม่อยากสร้ างเอง? ใช้ไฟล์สำเร็จรูป
ทุกไฟล์ใส่สูตรไว้ครบ ล็อกช่องสูตรกันพัง มีหน้าวิธีใช้ภาษาไทย เปิดแล้วกรอกข้อมูลได้เลย

การเงินส่วนตัว

ตาราง Excel รายรั บ-รายจ่ายรายเดือน
คำนวณผ่อนบ้าน / รถ / สินเชื่อ พร้ อมตารางผ่อน
วางแผนเกษียณ + คำนวณดอกเบี้ ยทบต้น
แผนปลดหนี้ หลายก้อน บัตรเครดิต / สินเชื่อ (Snowball vs
Avalanche)
สรุปค่าผ่อนทั้งหมด: มือถือ / รถ / บ้าน / บัตรเครดิต — เดือนนี้ จ่าย
เท่าไร ผ่อนหมดเมื่อไร
บันทึกทรั พย์สิน-หนี้ สิน: มีทั้งหมดเท่าไร ลบหนี้ แล้วเหลือเท่าไร
อัปเดตรายเดือน รายปี

ร้ านค้า / SME

Excel สต๊อกสินค้า รั บเข้า-จ่ายออก-คงเหลือ
Excel ขายหน้าร้ าน POS ยอดขายรายวัน ตัดสต๊อกอัตโนมัติ
ใบเสนอราคา / ใบแจ้งหนี้  / ใบเสร็จรั บเงิน (VAT 7%)
ใบกำกับภาษีเต็มรูป + ทะเบียนเลขที่รั นอัตโนมัติ + รายงานภาษีขาย
ลูกหนี้  / เจ้าหนี้  ติดตามยอดค้างและเครดิต + อายุหนี้

บัญชีรายรั บ-รายจ่ายร้ านค้า + งบกำไรขาดทุน + ประมาณการภาษี
เงินได้
ออเดอร์ แม่ค้าออนไลน์ + กำไรสุทธิหลังหักค่าธรรมเนียม
คำนวณต้นทุนเมนูอาหาร + ตั้งราคาขาย

HR / เงินเดือน

คำนวณเงินเดือน + ประกันสังคม + ภาษีหัก ณ ที่จ่าย พร้ อมสลิป
บันทึกวันลาพนักงาน + สิทธิ์ลาคงเหลือ (ลาป่ วย / ลากิจ / พักร้ อน)
ประเมินผลงานพนักงาน KPI + สมรรถนะ → เกรด ขึ้นเงินเดือน และ
โบนัส
คำนวณภาษีเงินได้บุคคลธรรมดา (มนุษย์เงินเดือน/ฟรี แลนซ์)

ครู / โรงเรี ยน

บันทึกคะแนน + ตัดเกรดนักเรี ยนอัตโนมัติ
เช็คชื่อนักเรี ยน มา / ขาด / ลา / สาย ทั้งเทอม
ติดตามการส่งงานนักเรี ยน + รวมคะแนนเก็บ รายคาบ

ขาย / จัดซื้อ

ยอดขายเทียบเป้ า + ค่าคอมมิชชั่นทีมขาย (รายคน รายเดือน)
ใบขอซื้อ (PR) + ผู้ อนุมัติตามวงเงิน + ติดตามสถานะจนออก PO

งานออฟฟิศ / โปรเจกต์

แผนงานโปรเจกต์ Gantt Chart + ติดตามความคืบหน้า

งานแต่ง / ชีวิต

วางแผนงานแต่ง: งบประมาณ + รายชื่อแขก + ซองช่วยงาน

